

Analiza matematyczna

Zajęcia nr 2

Wykres punktowy


`? ListPlot`

`ListPlot[{y1, y2, ...}]` plots points corresponding to a list of values, assumed to correspond to x coordinates 1, 2,
`ListPlot[{x1, y1}, {x2, y2}, ...]` plots a list of points with specified x and y coordinates.
`ListPlot[{list1, list2, ...}]` plots several lists of points. >>

Lista liczb:

```
punkty := {1.2, 4.4, 5., 2.4, 3.3, 2.4, 7.7, 10.2, 9.8}
```

```
ListPlot[punkty, PlotStyle -> PointSize[0.02],  
PlotLabel -> "Wykres punktowy", AxesLabel -> {"x", "y"}]
```




Argumenty to kolejne cyfry 1, 2, 3, ...

Aby podać wartości x -ów i y -ów trzeba stworzyć tablicę par liczb:


```
wspolrzedne := {{0, 1.2}, {0.2, 4.4}, {0.4, 5.},  
{0.7, 2.4}, {1.2, 3.3}, {2.5, 2.4}, {3.5, 7.7}, {5.1, 10.2}}
```

```
ListPlot[wspolrzedne, PlotStyle -> PointSize[0.02],
  PlotLabel -> "Wykres punktowy", AxesLabel -> {"x", "y"}]
```




Punkty połączone prostymi

```
ListPlot[wspolrzedne, PlotStyle -> PointSize[0.02],
  PlotLabel -> "ListPlot, Joined->True", AxesLabel -> {"x", "y"}, Joined -> True
]
```




```
ListLinePlot[wspolrzedne,
  PlotLabel -> "ListLinePlot", AxesLabel -> {"x", "y"}]
```



? InterpolationOrder

InterpolationOrder is an option for Interpolation, as well as ListLinePlot, ListPlot3D, ListContourPlot, and related functions, that specifies what order of interpolation to use. >>

```
{ListLinePlot[wspolrzedne, InterpolationOrder → 0],
 ListLinePlot[wspolrzedne, InterpolationOrder → 1],
 ListLinePlot[wspolrzedne, InterpolationOrder → 2]}
```



Krzywa regresji

? LinearModelFit

LinearModelFit[{y₁, y₂, ...}, {f₁, f₂, ...}, x] constructs a linear model of the form $\beta_0 + \beta_1 f_1 + \beta_2 f_2 + \dots$ that fits the y_i for successive x values 1, 2,

LinearModelFit[{x₁₁, x₁₂, ..., y₁}, {x₂₁, x₂₂, ..., y₂}, ..., {f₁, f₂, ...}, {x₁, x₂, ...}] constructs a linear model of the form $\beta_0 + \beta_1 f_1 + \beta_2 f_2 + \dots$ where the f_i depend on the variables x_k.

LinearModelFit[m, v] constructs a linear model from the design matrix m and response vector v. >>

LinearModelFit[{y₁, y₂, ...}, {f₁, f₂, ...}, x] znajduje parametry dopasowania funkcji $\beta_0 + \beta_1 f_1 + \beta_2 f_2 + \dots$ do danych y_i

Chcemy znaleźć parametry prostej regresji postaci $\beta_0 + \beta_1 x$

```
LMF = LinearModelFit[wspolrzedne, x, x]
```

```
FittedModel[2.31484 + 1.32951 x]
```


```
LMF[x]
```

```
2.31484 + 1.32951 x
```

```
LMF["ParameterTable"]
```

	Estimate	Standard Error	t-Statistic	P-Value
1	2.31484	0.964577	2.39985	0.0533059
x	1.32951	0.399487	3.32804	0.0158453

```
Show[
  ListPlot[wspolrzedne, PlotStyle -> PointSize[0.02]],
  Plot[LMF[x], {x, 0, 5}],
  PlotLabel -> "Regresja liniowa", AxesLabel -> {"x", "y"}
]
```



? NonlinearModelFit

`NonlinearModelFit[{y1, y2, ...}, form, {β1, ...}, x]` constructs a nonlinear model with structure *form* that fits the *y_i* for successive *x* values 1, 2, ... using the parameters β₁,

`NonlinearModelFit[{x11, x12, ..., y1}, {x21, x22, ..., y2}, ..., form, {β1, ...}, {x1, ...}]` constructs a nonlinear model where *form* depends on the variables *x_k*.

`NonlinearModelFit[data, form, cons, {β1, ...}, {x1, ...}]` constructs a nonlinear model subject to the parameter constraints *cons*. >>

`NonlinearModelFit[{y1, y2, ...}, form, {β1, ...}, x]` znajduje parametry dopasowania funkcji *form* zależnej od parametrów { β₁, β₂, ... } do danych *y_i*


```
NMF = NonlinearModelFit[wspolrzedne, a + b x + c x^2, {a, b, c}, x]
```

```
FittedModel[ 3.23429 - 0.519599 x + 0.38148 x^2 ]
```

```
NMF[x]
```

```
3.23429 - 0.519599 x + 0.38148 x^2
```

```
Show[
  ListPlot[wspolrzedne, PlotStyle -> PointSize[0.02]],
  Plot[NMF[x], {x, 0, 5}],
  PlotLabel -> "Regresja a+b x+c x^2", AxesLabel -> {"x", "y"}
]
```



```
NMF["ParameterTable"]
```

	Estimate	Standard Error	t-Statistic	P-Value
a	3.23429	1.11168	2.90938	0.0334239
b	-0.519599	1.37428	-0.378087	0.720886
c	0.38148	0.272989	1.39742	0.221132

Słupki błędów

Trzeba załadować pakiet `ErrorBarPlots``

```
Needs["ErrorBarPlots`"]
```


? ErrorBar

`ErrorBar[{negerror, poserror}]` represents error in the positive and negative directions.
`ErrorBar[yerr]` represents error *yerr* in both the positive and negative directions.
`ErrorBar[xerr, yerr]` represents errors specified for both the *x* and the *y* coordinates. >>

Współrzędne z słupkami błędów

```
wspolrzedneError := {
  {{0, 1.2}, ErrorBar[0.3, 0.5]},
  {{0.2, 4.4}, ErrorBar[0.5, 0.7]},
  {{0.7, 2.4}, ErrorBar[0.3, 0.5]},
  {{1.2, 3.3}, ErrorBar[0.4, 0.9]},
  {{2.5, 2.4}, ErrorBar[0.3, 0.8]},
  {{3.5, 7.7}, ErrorBar[0.2, 0.6]},
  {{5.1, 10.2}, ErrorBar[0.2, 0.5]}}
```


```
ErrorListPlot[wspolrzedneError,
  PlotLabel → "Słupki błędów", AxesLabel → {"x", "y"}]
```




PlotMarkers

Znaczniki danych

```
ListPlot[{wspolrzedne, punkty, {1, 4, 6, 3, 7, 3}},
  PlotLabel -> "3 serie danych", AxesLabel -> {"x", "y"},
  PlotLegends -> {"pierwszy", "drugi", "trzeci"},
  PlotStyle -> PointSize[0.03]]
```



```
ListPlot[{wspolrzedne, punkty, {1, 4, 6, 3, 7, 3}},
  PlotLabel -> "3 serie danych. PlotMarkers->Automatic",
  AxesLabel -> {"x", "y"}, PlotLegends -> {"pierwszy", "drugi", "trzeci"},
  PlotMarkers -> {Automatic, 15}]
```



Własny wybór znaczników danych

- ▲ `\[FilledUpTriangle]` (bez spacji)
- `\[FilledSquare]`
- `\[FilledCircle]`

```
ListPlot[{wspolrzedne, punkty, {1, 4, 6, 3, 7, 3}},
  PlotLegends → {"pierwszy", "drugi", "trzeci"},
  PlotLabel → "Własny wybór znaczników danych", AxesLabel → {"x", "y"},
  PlotMarkers → {"▲", "■", "●"},
  PlotStyle → {Green, Blue, Black}
]
```




Własny wybór znaczników danych - Graphics




? Graphics

Graphics[*primitives* , *options*] represents a two-dimensional graphical image. >>



```
ListPlot[{wspolrzedne, punkty, {1, 4, 6, 3, 7, 3}},
  PlotLegends → {"pierwszy", "drugi", "trzeci"},
  PlotLabel → "Własny wybór znaczników danych", AxesLabel → {"x", "y"},
  PlotMarkers → {
    {Graphics[{Green, Polygon[{{1, 0}, {0, Sqrt[3]}, {-1, 0}]}]}, 0.04},
    {Graphics[{Blue, Rectangle[]}]}, 0.04},
    {Graphics[{Black, Disk[]}]}, 0.04}
  ]
]
```



```
trojkat = Graphics[{Green, Polygon[{{1, 0}, {0, Sqrt[3]}, {-1, 0}]}]}
kwadrat = Graphics[{Blue, Rectangle[]}]
kolo = Graphics[{Black, Disk[]}]
okrag = Graphics[{Red, Circle[]}]
```



```
pb = Graphics[{Thickness[0.02],
  Red, Disk[{0, 0}, 1, {0, Pi}],
  Black, Circle[{0, 0}, 1],
  White, Disk[{-0.4, 0.1}, 0.2],
  White, Disk[{0.4, 0.1}, 0.2],
  Black, Circle[{-0.4, 0.1}, 0.2],
  Black, Circle[{0.4, 0.1}, 0.2]
}]
```



Generowanidist

? Table

`Table[expr, n]` generates a list of n copies of `expr`.

`Table[expr, {i, imax}]` generates a list of the values of `expr` when i runs from 1 to i_{max} .

`Table[expr, {i, imin, imax}]` starts with $i = i_{min}$.

`Table[expr, {i, imin, imax, di}]` uses steps di .

`Table[expr, {i, {i1, i2, ...}}]` uses the successive values i_1, i_2, \dots .

`Table[expr, {i, imin, imax}, {j, jmin, jmax}, ...]` gives a nested list. The list associated with i is outermost. >>

```
Table[n^2, {n, 6}]
```

```
{1, 4, 9, 16, 25, 36}
```

```
Table[n^2, {n, 0, 6}]
```

```
{0, 1, 4, 9, 16, 25, 36}
```


```
Table[n^2, {n, 0, 6, 2}]
```

```
{0, 4, 16, 36}
```

```
Table[{2 n, Sin[n]}, {n, 0, 4}]
```

```
{{0, 0}, {2, Sin[1]}, {4, Sin[2]}, {6, Sin[3]}, {8, Sin[4]}}
```

```
ListPlot[Table[{Cos[n], Sin[2 n]}, {n, 1, 360}],
  PlotLabel -> "{Cos[n], Sin[2n]}", AxesLabel -> {"x", "y"}]
```



Zadanie

1. Narysować wykres punktowy dla listy par liczb $\{x_i, y_i\}$ takich że
 x_i to cyfry od 0 do 30 co 2
 y_i to liczby postaci $\log(3 x_i + 7) + \text{rand}_i$
gdzie rand_i to liczba losowa z przedziału od 0 do 0.5 ($\text{RandomReal}[\{y_{\min}, y_{\max}\}]$)
2. Niech znaczniki to jakiś stworzony przez siebie znaczek przy użyciu Graphics
3. Dodać funkcję regresji logarytmicznej $\log(a+b x)$
4. Dodać tytuł i nazwy osi

? RandomReal

RandomReal[] gives a pseudorandom real number in the range 0 to 1.
RandomReal[{ x_{\min} , x_{\max} }] gives a pseudorandom real number in the range x_{\min} to x_{\max} .
RandomReal[x_{\max}] gives a pseudorandom real number in the range 0 to x_{\max} .
RandomReal[range, n] gives a list of n pseudorandom reals.
RandomReal[range, { n_1, n_2, \dots }] gives an $n_1 \times n_2 \times \dots$ array of pseudorandom reals. >>

```
RandomReal[{0, 0.5}]
```

```
0.103774
```